Data Management Techniques

Explanation:

So far we have looked out how to make tables and queries in access by using a graphical interface to help us.  All of our tables that we have made, or queries have had predefined models to help us aid our design.  However this section focuses on how to make tables or queries in Access “on the fly” so to say.  We will focus now on what the book calls QueryDefs and TableDefs.  Keep in mind that these QueryDefs and TableDefs are objects.  Therefore they have properties and methods.  For example, a given table “tblCustomers” is an object.  It’s properties are it’s attributes (fields) and it’s methods could be adding new records, adding new attributes (fields), and so on.  First let’s look at how to make a simple QueryDef.

Exercise1:

Create a new database.  Create a table called ‘tblWorkers’.  Create a primary key for the worker ID called ‘PK’, a field called ‘Hours’ which will hold the number of hours a week they work, and a field called ‘Wage’ which is their hourly wage, in currency format.  Populate the database with some dummy data.  Create different hours for workers…40, 50, 10, etc.  Add a module called ‘QandTDefs’.  Which means Query and Table Defs.  First let’s just see how to make a simple query during runtime.

Public Function FirstQueryDef(strQueryName As String, strSQL As String) As Boolean
Dim qdf As QueryDef

If strSQL = "" Then Exit Function
Set qdf = CurrentDb.CreateQueryDef(strQueryName)

qdf.SQL = strSQL

qdf.Close

RefreshDatabaseWindow

FirstQueryDef = True
End Function  

This function has two arguments.  The first is what you want to name the query, the second is what you want the SQL string to be.  The function returns a Boolean value...true if strSQL is not empty or false if it is empty.   First we define the QueryDef object.  This is just like saying, “I am going to make a query.”

Dim qdf As QueryDef

Next, we make sure that the SQL string that was sent to the function is not empty.  If it is then exit the function, and a default of ‘false’ will be returned from the function.

If strSQL = "" Then Exit Function

Next we say what the name of the query is.  Remember strQueryName is one of the arguments of the function.

Set qdf = CurrentDb.CreateQueryDef(strQueryName)

Then we define what the SQL source for the QueryDef is.  Since ‘qdf’ is an object, one of it’s properties is that it has an SQL string.

qdf.SQL = strSQL

Finally we close the QueryDef, and refresh the database window so we can see our new query.  Then we return a value of ‘true’ for the function.

qdf.Close

RefreshDatabaseWindow

FirstQueryDef = True

Now you are ready to create the query.  Let’s call the query qryFullTime, which will give us employees who work 40 or more hours a week.  Type this in the immediate window.

?FirstQueryDef("qryFullTime","SELECT * FROM tblWorkers WHERE Hours >= 40")

You should see a value of ‘True’ in your immediate window.  Now go to view your queries for this database.  You should see a new one called qryFullTime.  Now open it up.  Assuming you had workers with hours greater than or equal to 40, you should see a list of full time employees.  

Exercise 2:

Now that we have done this, go ahead and delete your query and delete your table (tblWorkers)…but keep your code.  Now we are going to see how to make your table at run time.  Create a new subroutine called FirstTableDef in the same module where FirstQueryDef is.  Copy the following code into your subroutine:

Public Sub FirstTableDef(strTableName As String)

Dim db As Database

Dim tbl As TableDef

Dim fld As Field

Dim idx As Index

Set db = CurrentDb()

Set tbl = db.CreateTableDef(strTableName)

Set fld = tbl.CreateField("PK", dbLong)

fld.OrdinalPosition = 1

fld.Attributes = dbAutoIncrField

tbl.Fields.Append fld

Set fld = tbl.CreateField("Hours", dbLong)

fld.OrdinalPosition = 2

tbl.Fields.Append fld

Set fld = tbl.CreateField("Wage", dbCurrency)

fld.OrdinalPosition = 3

tbl.Fields.Append fld

Set idx = tbl.CreateIndex("PrimaryKey")

idx.Primary = True
idx.Required = True
idx.Unique = True
Set fld = idx.CreateField("PK")

idx.Fields.Append fld

tbl.Indexes.Append idx

db.TableDefs.Append tbl

RefreshDatabaseWindow

End Sub 

This seems like a lot of code, but it is actually pretty simple and very logical.  First we declare our objects.

Dim db As Database

Dim tbl As TableDef

Dim fld As Field

Dim idx As Index

Then we set our database equal to the current database, then we create the TableDef (which is essentially creating the table) using the name that we passed via the one argument of the subroutine:

Set db = CurrentDb()

Set tbl = db.CreateTableDef(strTableName)

Next we create the properties for the first field in the database.  We create a field called “PK”, that has a data type dbLong.

Set fld = tbl.CreateField("PK", dbLong)

Next we set the position of the field (position 1), and specify an attribute about the field that says it is an autonumber.

fld.OrdinalPosition = 1

fld.Attributes = dbAutoIncrField

Then we add the field to the table

tbl.Fields.Append fld

Next we create another field that holds the number of hours each employee works.

Set fld = tbl.CreateField("Hours", dbLong)

fld.OrdinalPosition = 2

tbl.Fields.Append fld

Then we create the third field that holds the hourly wage of each employee.

Set fld = tbl.CreateField("Wage", dbCurrency)

fld.OrdinalPosition = 3

tbl.Fields.Append fld

Finally we must now say what the primary key of the table is.  To do this we create the index object, and define it as a primary key.

Set idx = tbl.CreateIndex("PrimaryKey")

idx.Primary = True
idx.Required = True
idx.Unique = True
Set fld = idx.CreateField("PK")

idx.Fields.Append fld

tbl.Indexes.Append idx

Finally we have added all of the needed fields to the table.  We now must add the table to the database schema.

db.TableDefs.Append tbl

RefreshDatabaseWindow

In the immediate window, run this line:

FirstTableDef("tblWorkers")

After running this, go to view all of your tables.  You should now see the table ‘tblWorkers’ which is identical to the table we made before.

Exercise 3:

Now that we have seen how to make both QueryDefs and TableDefs, you should be able to perform the following.

1. Delete all tables and querys from the current database.

2. Using the code above, create a table ‘tblWorkers’ once the database is opens.

3. Using recordset programming, add 5 new employees of your choice to the table.  Their wages and hours should be different.

4. Create a query using the QueryDef method that will get the full time employees (employees that work 40+ hours a week) and a query that will get all part time employees.

Steps for doing this:

1. Create a new subroutine that does each task

2. Make the first task a call to the subroutine FirstTableDef.

3. Make the second task a call to the subroutine that creates a record set to update table tblWorkers.

4. Make a third task a call to the subroutine (or function) that makes the full time employee query.

5. Make a fourth task a call to the subroutine (or function) that makes the part time employee query.

Exercise 4:
Step1:

Create a table called ‘tblTypes.’  Create a primary key auto number, then the second field should be called ‘Types’ (text field).  Put in these fields: dbCurrency, dbNumber, dbText. 

Step2:
Create a form called ‘frmTable’.  It should have one text field called txtTableName.  Then create a button called cmdMakeTable.  It should look like this.


[image: image1.wmf] 


Step3:

Create a form called frmFields that has a text field called txtFieldName, a combo box called cboTypes, a button named cmdAddField, and a button named cmdFinish.  It should look like this:


[image: image2.wmf] 


Step4:

The combo box should be based on a query that gets the types from ‘tblTypes.’  So the values in the combo box should look like this:


[image: image3.wmf] 


Tasks:

1. When the user enters a table name, then clicks ‘Make Table,’ they should be taken to frmFields.  They can add as many fields as they wish, then when they click Finish the table should be formed.

2. For simplistic purposes make the primary key a default to something like “PK” and set it’s ordinal position at 1.  Basically the user can not enter a primary key.  So when you are designing your code, make the first field “PK” with ordinal position 1, THEN start accepting the input the user gives you.

3. As the user enters new fields, make sure to increase the ordinal position of each field.  Also, do not let the user name two fields the same name.  You can check for this, by storing the already used field names into an array, then as they add a new field, check to see if it matches any of the names in the array.  Do not let the user enter a field without a name.

Email me any questions: tfad@mail.utexas.edu

